
15
Применение 
формул массивов

В ЭТОЙ ГЛАВЕ …

  Одноячеечные формулы массивов
  Многоячеечные формулы массивов

В предыд ущей главе были введены понятия массива и формулы массива,  а также даны 
соответствующие примеры. Настоящая глава посвящена формулам массивов и содержит 
полезные примеры, которые продемонстрируют их эффективность.

Эти примеры вы можете непосредственно использовать в работе, только в ряде случаев 
следует изменить имена диапазонов или ссылок. Во многие примеры легко внести измене-
ния, чтобы они работали именно так, как нужно для решения практических задач.

Все примеры этой главы содержатся в русифицированном файле Chapter15_RUS.
xlsx (названия листов соответствуют номерам рисунков), а также в нескольких 
авторских файлах, о которых будет сказано дополнительно.

Одноячеечные формулы массивов

Как описывалось в предыдущей главе, формулу массива можно вводить в одну ячейку 
(а не в диапазон ячеек). Такие формулы обрабатывают массивы, хранящиеся либо в диапа-
зоне ячеек, либо в памяти Excel. В этом разделе представлены некоторые дополнительные 
примеры одноячеечных формул массива.

О ПРИМЕРАХ ЭТОЙ ГЛАВЫ

Данная глава содержит ряд примеров формул массива. Помните, что формулы масси-
ва вводятся с  помощью комбинации клавиш <Ctrl+Shift+Enter>, а  не просто клавиши 
<Enter>. После нажатия этой комбинации клавиш Excel вставит введенную вами форму-
лу в фигурные скобки. Приведенные в тексте главы формулы массива заключены в фи-
гурные скобки только для наглядности, вам не придется вводить их в строке формул.

Все примеры этого раздела в авторском варианте содержатся в файле single-
cell array formulas.xlsx.

Суммирование в диапазоне, содержащем ошибки

Многие пользователи Excel знают, что функция СУММ не работает, если просуммиро-
вать диапазон ячеек, содержащих сообщения об ошибках (например, #ДЕЛ/0! или #Н/Д). 
На рис. 15.1 показан такой случай. Формула СУММ в ячейке С11 возвращает сообщение 
об ошибке, так как в диапазоне ячеек C4:C10, которые она складывает, содержатся со-
общения об ошибках.

Рис. 15.1. Формула массива может складывать 
значения в диапазоне, содержащем ошибки

Следующая формула массива возвращает сумму значений, даже если входной диапазон 
содержит сообщения об ошибке:
{=СУММ(ЕСЛИОШИБКА(C4:C10;"")}

Эта формула создает новый массив, включающий значения диапазона, но без сообще-
ний об ошибке. Функция ЕСЛИ заменяет сообщения об ошибке пустыми значениями. Затем 
функция СУММ складывает элементы “отфильтрованного” таким образом массива. Анало-
гичную методику можно использовать и для других функций, например МИН и МАКС.

В данном примере функция СУММ обрабатывает следующий внутренний массив, хра-
нящийся в памяти:
{8;20;12;"";"";5;10}

Возникают случаи, когда вместо ЕСЛИОШИБКА необходимо использовать другие 
функции. Дело в том, что ЕСЛИОШИБКА возвращает значение ИСТИНА при любых зна-
чениях ошибки (#Н/Д, #ЗНАЧ!, #ССЫЛКА!, #ДЕЛ/0!, #ЧИСЛО!, #ИМЯ? или #ПУСТО!). 
Функция ЕОШ возвращает значение ИСТИНА при любых значениях ошибки, кроме 
#Н/Д, а функция ЕНД — только при значениях ошибки #Н/Д.

Функция АГРЕГАТ предоставляет еще один способ суммирования диапазона, содер-
жащего ошибочные значения: 
=АГРЕГАТ(9,2,C4:C10)

Первый аргумент содержит значение 9 — код функции СУММ. Второй аргумент при-
казывает функции АГРЕГАТ игнорировать ошибки.

Подсчет количества ошибок в диапазоне

Представленная ниже формула массива подобна формуле предыдущего примера, за 
исключением того, что она возвращает количество ошибок в диапазоне Данные.
{=СУММ(ЕСЛИ(ЕОШИБКА(Данные);1;0))}

Эта формула создает массив, состоящий из единиц (если соответствующая ячейка со-
держит сообщение об ошибке) и нулей (если в ячейке отсутствует сообщение об ошибке).

Можно немного упростить формулу, удалив третий аргумент функции ЕСЛИ. В резуль-
тате, если условие не выполняется (ячейка не содержит сообщение об ошибке), функция 
ЕСЛИ будет возвращать значение ЛОЖЬ. Усовершенствованная формула возвращает такой 
же результат, что и предыдущая, но в ней нет третьего аргумента функции ЕСЛИ.
{=СУММ(ЕСЛИ(ЕОШИБКА(Данные);1))}

В действительности можно еще больше упростить эту формулу:
{=СУММ(ЕОШИБКА(Данные)*1)}

В этой версии формулы используются такие равенства:
ИСТИНА*1=1

и
ЛОЖЬ*1=0

Суммирование n наибольших значений в диапазоне

Следующая формула возвращает значение суммы десяти наибольших значений в диа-
пазоне Данные:
{=СУММ(НАИБОЛЬШИЙ(Данные;СТРОКА(ДВССЫЛ("1:10"))))}

Функция НАИБОЛЬШИЙ выполняется десять раз, причем каждый раз с другим вторым 
аргументом (1, 2, 3... 10). Результаты этих вычислений хранятся в новом массиве, который 
служит аргументом для функции суммирования.

Чтобы просуммировать другое количество значений, необходимо заменить число 10 
в аргументе функции ДВССЫЛ на другое значение.

Если количество суммируемых ячеек находится в ячейке C17, используйте оператор 
конкатенации для создания адреса диапазона для функции ДВССЫЛ.
{=СУММ(НАИБОЛЬШИЙ(Данные;СТРОКА(ДВССЫЛ("1:"&C17))))}

Для суммирования n наименьших значений в диапазоне следует заменить функцию 
НАИБОЛЬШИЙ функцией НАИМЕНЬШИЙ.

Вычисление среднего без учета нулевых значений

На рис. 15.2 показан рабочий лист, на котором вычисляется средний объем продаж. 
В ячейку B14 помещена следующая формула:
=СРЗНАЧ(Данные)

Рис. 15.2. Вычисление среднего значения, при котором 
исключаются ячейки, содержащие нуль

Эта формула вычисляет среднее значение именованного диапазона Данные. Посколь-
ку два продавца отсутствовали на работе некоторое время, в результате представленная 
выше формула неточно отображает средний уровень продаж торговых представителей.

Функция СРЗНАЧ не учитывает пустые ячейки, зато принимает в расчет ячейки, со-
держащие нулевые значения.

Следующая формула массива возвращает среднее значение в диапазоне и при этом ис-
ключает ячейки, содержащие нулевые значения:
{=СРЗНАЧ(ЕСЛИ(Данные<>0;Данные))}

Эта формула создает новый массив, состоящий исключительно из ненулевых значений. 
Функция СРЗНАЧ принимает этот массив в качестве своего аргумента. Тот же результат 
можно получить при использовании обычной формулы (не формулы массива):
=СУММ(Данные)/СЧЁТЕСЛИ(Данные;"<>0")

Вначале рассчитывается сумма всех значений ячеек в диапазоне, а затем она делится 
на количество ненулевых ячеек диапазона, полученное с помощью функции СЧЁТЕСЛИ.

Поиск значения в диапазоне

Для того чтобы определить, присутствует ли определенное значение в диапазоне ячеек, 
необходимо выбрать команду ГлавнаяРедактированиеНайти и выделитьНайти 
(или нажать комбинацию клавиш <Ctrl+F>), после чего вручную задать параметры по-
иска. Можно поступить другим образом и решить эту задачу автоматически с помощью 
формулы массива.

На рис. 15.3 показан рабочий лист со списком имен в диапазоне A3:E22, которому 
присвоено имя СписокИмен. Формула массива в ячейке D1 ищет имя, введенное в ячейке 
C1, которой присвоено имя Имя. Если заданное имя присутствует в списке имен, формула 
выводит текст Найдено. В противном случае она отображает фразу Не найдено.

Рис. 15.3. Использование формулы массива 
для определения наличия значения в диапазоне

Формула в ячейке D1 имеет следующий вид:
{=ЕСЛИ(ИЛИ(Имя=СписокИмен);"Найдено";"Не найдено")}

Эта формула сравнивает Имя с каждым элементом в диапазоне СписокИмен. Она соз-
дает новый массив, который состоит из логических значений ИСТИНА или ЛОЖЬ. Функция 
ИЛИ возвращает значение ИСТИНА, если хотя бы одно значение в новом массиве — ИСТИ-
НА. Функция ЕСЛИ использует этот результат для определения возвращаемого сообщения.

Более простой вариант этой формулы представлен ниже. Эта формула возвращает зна-
чение ИСТИНА, если имя найдено, в противном случае — ЛОЖЬ.
{=ИЛИ(Имя=СписокИмен)}

Еще один подход предполагает использование функции СЧЁТЕСЛИ:
==Если(СЧЁТЕСЛИ(СписокИмен153;Имя)>0;"Найдено";"Не найдено")

Подсчет отличающихся значений в двух диапазонах

Представленная ниже формула массива сравнивает соответствующие значения в двух 
диапазонах (имеющих имена МоиДанные и ТвоиДанные) и возвращает количество раз-
личных элементов в этих двух диапазонах. Если содержимое двух диапазонов идентично, 
формула возвращает нуль.
{=СУММ(ЕСЛИ(МоиДанные=ТвоиДанные;0;1))}

Пример показан на рис. 15.4.

Два диапазона должны иметь одинаковые размер и ориентацию.

Данная формула создает новый массив такого же размера, как и сравниваемые диа-
пазоны. Функция ЕСЛИ заполняет этот новый массив нулями и единицами (нуль — если 
есть различие, единица — если нет различий в соответствующих ячейках). Функция СУММ 
возвращает сумму значений этого массива.

Рис. 15.4. Использование формулы массива для подсчета 
количества отличий между двумя диапазонами

Тот же результат можно получить, применив более простую формулу:
{=СУММ(1*(МоиДанные<>ТвоиДанные))}

Эта версия формулы базируется на следующих равенствах:
ИСТИНА*1=1

и
ЛОЖЬ*1=0

Местоположение максимального значения диапазона

Для определения номера строки, в которой находится максимальное значение одно-
мерного вертикального массива Данные, можно воспользоваться следующей формулой:
{=МИН(ЕСЛИ(Данные=МАКС(Данные);СТРОКА(Данные);""))}

Функция ЕСЛИ создает новый массив, соответствующий диапазону Данные. Если соот-
ветствующая ячейка содержит максимальное значение диапазона Данные, то в новый мас-
сив включается номер ее строки. В противном случае массив содержит пустую текстовую 

строку. Функция МИН использует этот массив как второй аргумент и возвращает мини-
мальное значение, которое соответствует номеру строки, в которой находится максималь-
ное значение диапазона Данные.

Если диапазон Данные содержит несколько ячеек с максимальным значением, то воз-
вращается номер первой такой ячейки.

Представленная ниже формула массива подобна предыдущей, но она возвращает дейст-
вительный адрес ячейки с максимальным значением в диапазоне Данные. В формуле ис-
пользуется функция АДРЕС, которая имеет два аргумента: номер строки и номер столбца.
{=АДРЕС(МИН(ЕСЛИ(Данные=МАКС(Данные);СТРОКА(Данные);""));
   СТОЛБЕЦ(Данные))}

Если диапазон Данные занимает больше одного столбца, следующая формула массива 
вернет правильный адрес:
{=АДРЕС(МИН(ЕСЛИ(Данные=МАКС(Данные);СТРОКА(Данные);""));
МИН(ЕСЛИ(Данные=МАКС(Данные);СТОЛБЕЦ(Данные);"")))}

Поиск номера строки, в которой находится n-е 
значение, совпадающее с заданным

Приведенная ниже формула ищет в одномерном вертикальном диапазоне Данные зна-
чения, которые совпадают с содержимым ячейки Значение, и возвращает номер строки 
n-го совпадающего элемента.
{=НАИМЕНЬШИЙ(ЕСЛИ(Данные=Значение;СТРОКА(Данные);"");n)}

Функция ЕСЛИ создает новый массив, состоящий из номеров строк, в которых значе-
ния из диапазона Данные совпадают со значением Значение. Если значение из диапазона 
Данные не совпадает со значением Значение, то новый массив содержит пустую тексто-
вую строку. Функция НАИМЕНЬШИЙ обрабатывает этот новый массив и возвращает n-й 
наименьший номер столбца.

Если n превышает количество значений в диапазоне Данные, равных Значение, или 
значение Значение не найдено, формула возвращает ошибку #ЧИСЛО!.

Получение самого длинного текста в диапазоне

Следующая формула отображает содержимое ячейки диапазона Данные, в которой 
находится наибольшее количество символов. Если самый длинный текст содержится в не-
скольких ячейках, то возвращается содержимое первой такой ячейки.
{=ИНДЕКС(Данные;ПОИСКПОЗ(МАКС(ДЛСТР(Данные));ДЛСТР(Данные);
ЛОЖЬ);1)}

Эта формула работает с двумя массивами, каждый из которых имеет длину соответ-
ствующего элемента из диапазона Данные. Функция МАКС находит наибольшее значение, 
которое соответствует элементу с самым длинным текстом. Функция ПОИСКПОЗ вычисля-
ет смещение ячейки с максимальной длиной. Функция ИНДЕКС возвращает содержимое 
ячейки, в которой находится текст с наибольшим количеством символов. Эта формула 
возвращает правильный результат только в том случае, если диапазон Данные состоит из 
одного столбца (рис. 15.5).

Рис. 15.5. Использование формулы массива для поиска 
самого длинного текста в диапазоне

Определение допустимых значений диапазона

Перед вами может возникнуть следующая задача: проконтролировать содержимое спи-
ска на предмет допустимости в нем значений. Предположим, что список всех допустимых 
чисел хранится в диапазоне Основа. При импорте списка чисел в диапазон МойСписок 
нужно убедиться, что все числа допустимы. Для этого необходимо сравнить элементы 
в импортируемом списке с элементами в списке Основа (рис. 15.6).

Рис. 15.6. Использование формулы массива для подсчета 
и идентификации элементов, отсутствующих в списке

Приведенная ниже формула массива возвращает значение ИСТИНА, если каждый эле-
мент диапазона МойСписок найден в диапазоне Основа. Оба этих диапазона должны со-
стоять из одного столбца (это не зависит от числа строк):
{=ЕНД(ПОИСКПОЗ(ИСТИНА;ЕНД(ПОИСКПОЗ(МойСписок;Основа;0));0))}

Можно также определить количество недопустимых значений. Иными словами, если 
требуется подсчитать количество элементов в диапазоне МойСписок, которые отсутству-
ют в диапазоне Основа, воспользуйтесь формулой
{=СУММ(1*ЕНД(ПОИСКПОЗ(МоиСписок;Основа;0)))}

Для определения первого недопустимого элемента в диапазоне МойСписок используй-
те формулу
{=ИНДЕКС(МойСписок;ПОИСКПОЗ(ИСТИНА;
   ЕНД(ПОИСКПОЗ(МойСписок;Основа;0));0))}

Вычисление суммы цифр числа

Я не могу представить себе, зачем это может понадобиться на практике, но это хорошая 
демонстрация возможностей формул массивов. Для вычисления суммы цифр в положитель-
ном целом числе, которое хранится в ячейке A1, можно использовать следующую формулу:
{=СУММ(ПСТР(A1;СТРОКА(ДВССЫЛ("1:"&ДЛСТР(A1)));1)*1)}

Например, если ячейка A1 содержит значение 409, приведенная выше формула вернет 
значение 13 (4+0+9=13).

Чтобы понять, как работает эта формула, рассмотрим функцию СТРОКА в следующей 
формуле:
{=СТРОКА(ДВССЫЛ("1:"&ДЛСТР(A1)))}

Эта формула возвращает массив последовательных целых чисел, который начинается 
с единицы и заканчивается количеством цифр в числе, находящемся в ячейке A1. Напри-
мер, если ячейка A1 содержит значение 409, функция ДЛСТР вернет значение 3, а функцией 
СТРОКА будет сгенерирован следующий массив:
{1;2;3}

Подробная информация об использовании функции ДВССЫЛ приведена в главе 14.

Полученный массив впоследствии используется в качестве второго аргумента функции 
ПСТР. Если в формулу с ПСТР подставить полученные в описанном примере значения, она 
будет упрощена:
{ПСТР(409;{1;2;3};1)*1}

Эта формула генерирует массив, состоящий из трех элементов:
{4;0;9}

В результате общая формула c функцией СУММ станет более простой и будет иметь 
следующий вид:
=СУММ({4;0;9})

Результат вычислений равен 13.

Функция ПСТР возвращает текстовую строку, поэтому при преобразовании текста 
в число элементы массива, находящиеся в аргументе функции ПСТР, умножаются 
на единицу. Альтернативный путь — использование функции ЗНАЧЕН, которая также 
преобразует строку текста, отображающую число, в числовое значение.

Заметьте, что описанная выше формула не ра-
ботает с отрицательными значениями, так как знак 
“минус” не является цифрой. Ниже приводится 
формула, в которой исправлен данный недостаток. 
Для этого используется функция ABS, которая 
возвращает абсолютное значение числа. На рис. 15.7 
представлен рабочий лист, содержащий в ячейке B4 
следующую формулу:
{=СУММ(ЗНАЧЕН(ПСТР(ABS(A2);СТРОКА(ДВССЫЛ(
"1:"&ДЛСТР(ABS(A2))));1)))}

Формула в ячейке B4 была скопирована в другие 
ячейки столбца B с целью вычислить суммы цифр 
в числах, находящихся в столбце A.

Суммирование округленных значений

 На рис. 15.8 показан рабочий лист, который демонстрирует характерную проблему всех 
электронных таблиц — ошибку округления. Как видно на рисунке, общая сумма в ячейке 
E7 не является точной суммой чисел в столбце E. Значения в столбце E используют чис-
ловой формат, в котором отображаются две цифры после запятой. Фактические значения 
имеют больше знаков после запятой, и некоторые из них не отображаются. Результирую-
щее влияние погрешностей округления может привести к общему неточному результату. 
Общее значение составляет $168,320997, отображаемое — $168,32.

Рис. 15.8. Использование формулы массива устраняет ошибку округления

Следующая формула массива создает новый массив, который состоит из значений 
столбца E, округленных до двух десятичных знаков:
{=СУММ(ОКРУГЛ(E2:E4;2))}

Эта формула возвращает правильную величину суммы округленных значений  — 
$168,31.

Для устранения таких ошибок округления можно также ввести в ячейки столбца E 
уже округленные значения с помощью функции ОКРУГЛ. Этот подход не требует создания 
формулы массива.

Для суммирования каждого n-го элемента диапазона используется следующая формула:
{=СУММ(ЕСЛИ(ОСТАТ(СТРОКА(ДВССЫЛ("1:"&
ЧСТРОК(Данные156)))-1;n)=0;Данные156;""))}

Рис. 15.7. Формула массива вычисляет 
сумму цифр любого целого числа

Суммирование каждого n-го значения в массиве

Предположим, что существует диапазон значений, в котором необходимо определить 
сумму каждого третьего значения: первого, четвертого, седьмого и т.д. Один из способов 
решения этой задачи — внесение адреса ячеек в формулу суммирования. Однако лучшим 
решением будет использование формулы массива.

Обратимся к данным рис. 15.9. Значения хранятся в диапазоне B4:B22 с именем Дан-
ные156, а значение n задается в ячейке D4, названной n.

Рис. 15.9. Формула массива возвращает 
сумму каждого n-го значения в диапазоне

Эта формула создает массив последовательных целых чисел. Функция ОСТАТ прини-
мает его в качестве первого аргумента. Второй аргумент функции ОСТАТ — значение n. 
Функция ОСТАТ создает другой массив, состоящий из остатков деления каждого после-
довательного целого числа на n. При равенстве элемента этого массива нулю (это значит, 
что деление прошло без остатка) происходит суммирование соответствующих значений 
в диапазоне Данные.

Если принять значение n равным нулю, приведенная выше формула возвращает со-
общение об ошибке. Улучшенная формула массива включает функцию ЕСЛИ для успешной 
работы при n=0.
{=ЕСЛИ(n=0;0;СУММ(ЕСЛИ(ОСТАТ(СТРОКА(ДВССЫЛ("1:"
&ЧСТРОК(Данные156)))-1;n)=0;Данные156;"")))}

Эта формула работоспособна только в том случае, когда диапазон Данные состоит из 
одного столбца значений. Она не работает, если диапазон состоит из более чем одного 
столбца или даже одной строки.

Приведенная выше формула всегда включает первый элемент диапазона. Например, 
если n=5, будут суммироваться следующие элементы: 1, 6, 11 и 16. Рассмотрим несколько 
модифицированную версию формулы, которая начинает суммирование именно с n-го эле-
мента. Когда n=5, формула суммирует следующие элементы: 5, 10 и 15.
{=ЕСЛИ(n=0;0;СУММ(ЕСЛИ(ОСТАТ(СТРОКА(ДВССЫЛ("1:"
&ЧСТРОК(Данные156)))-n;n)=0;Данные156;"")))}

Для создания формулы, работающей с  горизонтальным диапазоном, необходимо 
транспонировать массив целых чисел, генерируемых функцией СТРОКА. Модифицирован-
ная формула массива, предназначенная для обработки горизонтального диапазона Дан-
ные, имеет следующий вид:
{=ЕСЛИ(n=0;0;СУММ(ЕСЛИ(ОСТАТ(ТРАНСП(СТРОКА(ДВССЫЛ("1:"
&ЧСТРОК(Данные))))-1;n)=0;Данные;"")))}

ПРОВЕРКА ПРАВИЛЬНОСТИ ВЫЧИСЛЕНИЯ ФОРМУЛ

Чтобы лучше понять, как выполняется конкретная сложная формула массивов, обра-
титесь к  специальному средству отладки формул. Для  этого выделите ячейку с  фор-
мулой и  выберите команду ФормулыЗависимости формулВычислить форму-
лу. На  экране откроется диалоговое окно Вычисление формулы, показанное ниже. 
Щелкните на кнопке Вычислить несколько раз, и вы увидите промежуточные резуль-
таты вычисления формулы. Это подобно замедленному, пошаговому воспроизведению 
процесса вычисления.

Удаление нечисловых символов из текстовой строки

Приведенная ниже формула извлекает числа из текстовой строки. Например, если при-
менить формулу к строке, содержащей текст ABC145Z, будет возвращено значение 145.
{=ПСТР(A1;ПОИСКПОЗ(0;(ЕОШИБКА(ПСТР(A1;
 СТРОКА(ДВССЫЛ("1:"&ДЛСТР(A1)));1)*1)*1);0);
 ДЛСТР(A1)-СУММ((ЕОШИБКА(ПСТР(A1;
 СТРОКА(ДВССЫЛ("1:"&ДЛСТР(A1)));1)*1)*1)))}

Эта формула работает только с одним внедренным числом. Например, она даст непра-
вильный результат при работе со строкой X45Z99 (вернет 45Z9).

Поиск ближайшего значения в диапазоне

Пусть ячейка с произвольным числом имеет имя Цель157. Для определения ближай-
шего к нему числа из массива Данные157 воспользуемся формулой массива:
{=ИНДЕКС(Данные157;ПОИСКПОЗ(НАИМЕНЬШИЙ(ABS(
Цель157-Данные157);1);ABS(Цель157-Данные157);0))}

Если в массиве существует два числа, ближайших к Цель157, то формула вернет зна-
чение первого из них. На рис. 15.10 представлен пример использования этой формулы.

Рис. 15.10. Формула массива возвращает ближайшее значение

В этом примере массив Данные157 находится в диапазоне A4:A22, значение Цель157 
составляет 45. Формула массива находится в ячейке D5 и возвращает ближайшее к значе-
нию Цель157 значение, равное 48.

Получение последнего значения в столбце

Предположим, существует рабочий лист, в столбец A которого периодически вносятся 
новые данные, и нам необходимо отобразить последнее введенное значение. Если стол-
бец A не содержит пустых ячеек, данную задачу можно решить сравнительно просто, не 
прибегая к использованию формул массива:
=СМЕЩ(A1;СЧЁТЗ(A:A)-1;0)

В этой формуле функция СЧЁТЗ используется для подсчета количества непустых ячеек 
в столбце A. Полученное значение (за вычетом единицы) используется в качестве второ-
го аргумента функции СМЕЩ. Например, пусть последнее значение было введено в строку 
с номером 100. Функция СЧЁТЗ вернет значение 100, а функция СМЕЩ — значение в ячей-
ке, находящейся на 99 строк ниже ячейки A1.

В то же время столбец A вполне может содержать хотя бы одну пустую ячейку в диа-
пазоне. Это приведет к тому, что предыдущая формула будет выполняться неправильно. 
Дело в том, что функция СЧЁТЗ не подсчитывает пустые ячейки.

Приведенная ниже формула массива возвращает содержимое последней непустой 
ячейки столбца A.
{=ИНДЕКС(A1:A;МАКС(СТРОКА(A1:A)*(A1:A<>"")))}

Естественно, что эту формулу можно модифицировать для работы с любым другим 
столбцом. Для этого вместо ссылок на столбец A введите ссылки на любой другой столбец.

Если в столбце есть ошибка, эта формула не работает.

Эту формулу нельзя размещать в  том столбце, который она обрабатывает. 
Игнорирование этого правила приводит к циклической ссылке. Однако можно не-
много модифицировать формулу. Например, разместите формулу в первой строке 
и измените ссылки так, чтобы они начинались со второй строки.

Существует также альтернативная формула (не формула массива), возвращающая 
последнее значение в столбце. Приведенная ниже формула возвращает последнюю 
непустую ячейку столбца A.
=ПРОСМОТР(2;1/(А:А<>"");A:A)

Единственная проблема состоит в том, что непонятно, как эта формула работает. 
Но она вполне надежная и, мне кажется, более эффективная, чем формула масси-
ва. От формулы массива она отличается тем, что игнорирует ошибочные значения. 
Фактически она возвращает последнюю непустую ячейку, не содержащую ошибки.

Получение последнего значения в строке

Следующая формула массива аналогична предыдущей, однако возвращает последнюю 
непустую ячейку строки (в данном случае строки 1):
{=ИНДЕКС(1:1;МАКС(СТОЛБЕЦ(1:1)*(1:1<>"")))}

Для использования этой формулы в другой строке замените ссылку 1:1.
На рис. 15.11 показан пример для последних значений в столбце и строке.

Рис. 15.11. Использование формулы массива для возвращения 
последней непустой ячейки в столбце или строке

Приведенная ниже альтернативная формула (не формула массива) возвращает послед-
нюю непустую и, кроме того, не ошибочную ячейку в строке.
=ПРОСМОТР(2;1/(1:1<>"");1:1)

Многоячеечные формулы массивов

В предыдущей главе вы узнали об использовании многоячеечных формул массива, воз-
вращающих несколько значений. В этой главе представлено еще несколько полезных при-
меров использования таких формул массива. Большинство приведенных формул возвра-
щает несколько или все значения диапазона, реорганизованные в определенном порядке.

Все примеры этого раздела содержатся в файлах multicell array formulas.
xlsx и Chapter15_RUS.xlsx.

При вводе многоячеечной формулы массива необходимо сначала выделить весть диа-
пазон, а затем ввести формулу и нажать комбинацию клавиш <Ctrl+Shift+Enter>.

Извлечение положительных значений из диапазона

Представленная ниже формула массива работает с одномерным вертикальным масси-
вом, имеющим имя Данные. Формула массива расположена в диапазоне такого же разме-
ра, что и диапазон Данные. Для извлечения из диапазона Данные только положительных 
значений (нулевые и отрицательные значения игнорируются) используется следующая 
формула массива:
{=ИНДЕКС(Данные;НАИМЕНЬШИЙ(ЕСЛИ(Данные>0;
  СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные))));
  СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные)))))}

Как видно на рис. 15.12, эта формула работает, но имеет некоторые недостатки. Диапа-
зон Данные находится в ячейках A5:A24, а формула массива введена в диапазон C5:C24. 
Однако для ячеек, которые не содержат значения, формула массива возвращает сообщение 
об ошибке #ЧИСЛО!.

Модифицированная формула массива, введенная в диапазон ячеек E5:E24, использует 
функцию ЕСЛИОШИБКА во избежание сообщений об ошибках.
{=ЕСЛИОШИБКА(ИНДЕКС(Данные;НАИМЕНЬШИЙ(ЕСЛИ(Данные>0;
  ДВССЫЛ(("1:"&ЧСТРОК(Данные))));
  СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные)))));"")}

Извлечение непустых ячеек из диапазона

Формула этого раздела — один из вариантов формулы предыдущего раздела. Она ра-
ботает с одномерным вертикальным массивом Данные. Формула массива вводится в та-
кой же по размерности диапазон, что и диапазон Данные, и возвращает только непустые 
ячейки из этого диапазона.
{=ЕСЛИОШИБКА(ИНДЕКС(Данные;
   НАИМЕНЬШИЙ(ЕСЛИ(Данные<>0;
   ДВССЫЛ(("1:"&ЧСТРОК(Данные))));
   СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные)))));"")}

В более старых версиях программы (до Excel 2007) можно использовать следующую 
формулу.

{=ЕСЛИ(ЕОШ(НАИМЕНЬШИЙ(ЕСЛИ(Данные<>"";
  СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные))));
  СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные)))));"";
  ИНДЕКС(Данные;НАИМЕНЬШИЙ(ЕСЛИ(Данные<>"";
  СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные))));
  СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные))))))}

Рис. 15.12. Использование формулы массива для возвращения 
из диапазона только положительных значений

Изменение порядка следования элементов 
в диапазоне на противоположный

Представленная ниже формула массива работает с одномерным вертикальным мас-
сивом Данные. Она расположена в диапазоне, имеющем такой же размер, что и диапазон 
Данные. Для возвращения массива с противоположным порядком следования элементов 
используется следующая формула.
{=ЕСЛИ(ИНДЕКС(Данные;ЧСТРОК(Данные)
  СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные)))+1)="";"";
  ИНДЕКС(Данные;ЧСТРОК(Данные)-
  СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные)))+1))}

Присвоим диапазону A4:A13 имя Данные, а формулу массива введем в диапазон 
C4:C13. На рис. 15.13 продемонстрирована работа представленной выше формулы массива.

Динамическая сортировка значений диапазона

 Предположим, существует рабочий лист с одномерным вертикальным диапазоном 
Данные. Для сортировки значений массива по убыванию воспользуемся следующей фор-
мулой массива. Следует заметить, что данная формула работает только с числовыми зна-
чениями, но не с текстом:
{=НАИБОЛЬШИЙ(Данные;СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные))))}

Рис. 15.13. Порядок следования элементов массива изменяется 
на противоположный с помощью формулы массива

На рис. 15.14 показан пример; диапазон ввода находится в столбце A; формула массива 
введена в столбец C.

Эта формула может оказаться полезной, если требуется отсортировать данные немед-
ленно. Начните с присвоения диапазону данных имени Данные, а затем введите формулу 
в другой диапазон данных, размещенный в тех же строках.

Если в исходной ячейке не содержится данных, формула массива отображает значе-
ние ошибки #ЧИСЛО!. В следующей, модифицированной формуле используется функция 
ЕСЛИОШИБКА. Она позволяет избежать отображения на экране сообщения об ошибке. Эта 
формула используется в столбце E на рис. 15.11.
{=ЕСЛИОШИБКА(НАИБОЛЬШИЙ(Данные;
   СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные))));"")}

Эта формула должна вводиться в такой же по размеру диапазон, что и диапазон Дан-
ные. Она тоже сортирует только числовые значения.

Для сортировки значений Данные по возрастанию можно использовать такую формулу:
{=НАИМЕНЬШИЙ(Данные;СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные))))}

Возвращение списка уникальных значений диапазона

Если существует одномерный диапазон Данные, то для вывода списка уникальных эле-
ментов этого диапазона можно воспользоваться следующей формулой.
{=ИНДЕКС(Данные;НАИМЕНЬШИЙ(ЕСЛИ(ПОИСКПОЗ(Данные;
   Данные;0)=СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные)));
   ПОИСКПОЗ(Данные;Данные;0);"");
   СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные)))))}

Эта формула не работает, если диапазон Данные содержит хотя бы одну пустую ячей-
ку. В незаполненных ячейках формулы массива будет отображаться сообщение об ошиб-
ке #ЧИСЛО!. На рис. 15.15 представлен пример применения формулы массива. Диапазон 
A5:A23 имеет имя Данные, а формула массива введена в диапазон C5:C23. Обратите вни-
мание на то, что в незаполненных ячейках отображается ошибка #ЧИСЛО!.

Рис. 15.14. Формула массива, введенная в столбце C, 
сортирует значения, введенные в столбец A

Чтобы избежать отображения значения ошибки, лучше воспользоваться функцией 
ЕСЛИОШИБКА.
{ЕСЛИОШИБКА(ИНДЕКС(Данные;НАИМЕНЬШИЙ(ЕСЛИ
   (ПОИСКПОЗ(Данные;Данные;0)=СТРОКА(ДВССЫЛ
   (""&ЧСТРОК(Данные)));
    ПОИСКПОЗ(Данные;Данные;0);"")))));"")}

На рис. 15.15 эта формула используется в столбце E.

Рис. 15.15. Формула массива применяется для создания 
списка уникальных элементов диапазона данных

Отображение календаря в диапазоне

На рис. 15.16 представлен результат одной из формул, отображающей “живой” кален-
дарь в диапазоне ячеек. Если изменить дату верхней части диапазона, календарь автомати-
чески отобразит даты соответствующего месяца и года.

Рабочая книги данного примера находится в файле array formula calendar.
xslx.

Рис. 15.16. Отображение календаря с помощью одной формулы массива

1. Отформатируйте диапазон B4:H9 с пользовательским форматом d. В таком фор-
мате на экране будут отображаться только даты. Для выбора этого формата вос-
пользуйтесь категорией Дополнительно диалогового окна форматирования ячеек.

2. Скорректируйте ширину столбцов согласно своим предпочтениям.

Изменение месяца и года в ячейке B2 приводит к автоматическому обновлению кален-
даря. После создания календаря можно скопировать его диапазон в любое место рабочего 
листа или книги.

Формула массива фактически возвращает даты, но ячейки отформатированы так, что-
бы отображать только дни. Также обратите внимание на то, что в формулах массива ис-
пользуются массивы констант.

Массивы констант рассматриваются в главе 14.

Рассмотрим еще одну версию приведенной выше формулы. Функция ЕСЛИ в исходной 
формуле проверяет каждую дату на принадлежность текущему месяцу. Если это не так, 
функция ЕСЛИ возвращает пустую строку.

Приведенную выше формулу массива можно несколько упростить, удалив функцию ЕСЛИ.
=ДАТА(ГОД(B2);МЕСЯЦ(B2);1)-(ДЕНЬНЕД(ДАТА(ГОД(B2);МЕСЯЦ(B2);
1))-1)+{0;1;2;3;4;5}*7+{1;2;3;4;5;6;7}-1

Эта версия формулы отображает также дни предыдущего и следующего месяцев.
На рис. 15.17 показаны двенадцать экземпляров календаря на основе формул массивов 

для всего года.

Рис. 15.17. Календарь на год, состоящий из формул массивов

